/*  生成一个提示信息窗口以便用时显示出来
    在右下角显示一个提示窗口
    默认的 tooltip 窗口太小了
    */

; 生成 tip 窗口的 gui 对象
_lzTipGui := Gui("AlwaysOnTop -Caption ToolWindow Disabled")
; 设置后续添加控件的文字大小及粗细
_lzTipGui.setFont("s24 w555", "Microsoft YaHei")
; 添加文本控件用来显示信息，设为右对齐
_lzTipTxtControl := _lzTipGui.add("Text"
        , "0x80 W" . A_ScreenWidth - 55 . " Right")

lzShowTip(txt := "", secs := 3, color := 0x33AA33) {
    ; 设置背景颜色以便穿透
    _lzTipGui.backColor := format("{1:x}", color ^ 0x1)
    ; 让背景颜色透明
    WinSetTransColor(_lzTipGui.backColor, _lzTipGui)
    ; 按参数修改文字颜色
    _lzTipTxtControl.opt("c" . format("{1:x}", color))
    ; 设置要显示的文字
    _lzTipTxtControl.text := txt
    if txt {
        _lzTipGui.show("NA y" . A_ScreenHeight - 128)
        setTimer(() => _lzTipGui.hide(), - secs * 1000)
    }
}
